
Modélisation ARIMA  
des dépenses de consommation des ménages français  

de 1978 :1 à 2002 :4 
 
 
La table SAS a.consoFra contient la série de consommation, nommée ‘conso’ et la variable 
‘date’, allant de 1978Q1 à 2002Q4, soit 100 observations. 
 
Création de la variable ‘lconso’ qui est le logarithme de la variable initiale, et de la variable ‘t’ 
qui correspond à une tendance : 
data a.consoFra; 
set a.consoFra; 
lconso=log(conso); 
t=_n_; 
run; 
 
Représentation graphique de la série : 
proc gplot data=a.consoFra; 
symbol i=join; 
plot conso*date; 
plot lconso*date; 
run; 
 

 
L’allure croissante de la série permet de conclure à la non stationnarité de la série. 
 
Analyse des autocorrélations simples : 
proc arima data=a.consoFra; 
identify var=lconso;run;quit; 

 
La décroissance lente des autocorrélations simples indiquent la présence de non stationnarité 
(DS ou TS). 

Calcul de la densité spectrale et représentation graphique : 
PROC SPECTRA DATA=a.consoFra CENTER OUT=a.spectre P S ; 
VAR lconso ; 
WEIGHT PARZEN ; 
RUN ; 
PROC GPLOT DATA=a.spectre ; 
PLOT S_01*FREQ; 
PLOT S_01*PERIOD ; 
RUN ; 
quit; 
 

  
Elle indique aussi la présence d’une composante tendancielle (non stationnaire). 
 
Caractérisation de la non-stationnarité : 
 
1) On régresse la série sur une tendance déterministe et on récupère le résidu, pour analyser 
(visuellement grâce aux autocorrélations) sa stationnarité : 
proc reg data=a.consoFra; 
model lconso=t; 
output out=a.resTS r=res p=fit; 
run;quit; 
proc arima data=a.resTS; 
identify var=res; 
run; 

 
Le fait de retirer une tendance semble avoir créé une autocorrélation très forte. 
 
2) On stationnarise par différenciation et on analyse (visuellement) sa stationnarité : 
proc arima data=a.consoFra; 
identify var=lconso(1);run; 

La série différenciée semble stationnaire mais sa moyenne n’est pas forcément nulle. On peut 
alors penser que la série de consommation (en log) est un processus I(1) avec dérive. Pour 
confirmer cette intuition, on mène les tests de Dickey et Fuller (appelé DF par la suite).  
Il faut pour cela choisir au préalable le nombre de retards à introduire dans la régression du 
test de DF pour blanchir les résidus. 
 
Au vu de la représentation des autocorrélations partielles de la série différenciée, on peut 
penser à un AR(2). On peut alors mener le test de DF pour p=2. 

 
 
proc arima data=a.consoFra; 
identify var=lconso stationarity=(adf=(2)) ; 
run; 

 
La p-value associée au test de DF dans le modèle M3 (Trend) (Pr<Tau=0.7097) conduit à 
accepter l’hypothèse nulle de racine unitaire. On teste alors conjointement la racine unitaire et 
la nullité du coefficient de la tendance par le test F3 afin de s’assurer que le modèle M3 était 
pertinent pour tester la racine unitaire. La pvalue de ce test (Pr>F=0.8) conduit à accepter 
l’hypothèse nulle de nullité jointe. Il faut donc refaire le test de DF dans le modèle M2 (Single 
Mean) : la pvalue du test DF (Pr<Tau=0.9865) conduit à accepter l’hypothèse nulle de racine 
unitaire. On teste alors conjointement la racine unitaire et la nullité du coefficient de la 
constante par le test F2 afin de s’assurer que le modèle M2 était pertinent pour tester la racine 
unitaire. La pvalue de ce test (Pr>F=0.001) conduit à rejeter l’hypothèse nulle de nullité 

jointe. Ainsi, on peut conclure que la consommation est I(1) avec dérive, c’est-à-dire qu’il y a 
une racine unitaire avec une constante significative pour sa différence première (soit son taux 
de croissance), ce qui signifie aussi que la consommation est non stationnaire de type 
stochastique (DS) et donc que son taux de croissance est stationnaire. 
 
Modélisation ARMA de la partie stationnaire, c’est-à-dire de (1-L)lconso : 
 
Création de la variable et représentation graphique : 
data a.dlconso;set a.consoFra; 
dlconso=dif(lconso); 
run; 
proc gplot data=a.dlconso; 
plot dlconso*t ; 
run; 

 
Identification du modèle ARMA(p,q): 
proc arima data=a.consoFra; 
identify var=lconso(1) nlag=12; 
run; 
L’option nlag=12 permet de ne calculer que 12 autocorrélations simples, inverses et partielles 
pour faciliter l’examen graphique. 

On peut hésiter entre un AR(2) et un MA(2). 
Afin d’obtenir le critère BIC pour différents modèles ARMA estimés, il suffit de mettre 
l’option minic. 
 
proc arima data=a.consoFra; 
identify var=lconso(1) minic;run; 

 
Le critère BIC nous conduit à choisir plutôt un AR(2). 
 
L’estimation de l’AR(2) est obtenue par la commande : 
estimate p=2;run; 
 
Le modèle AR(2) est alors estimé sur la variable définie dans la commande identify qui a 
été exécuté en dernier (ici lconso(1)). 

Si on avait voulu estimer un MA(2) : 
estimate q=2;run; 
 
Pour estimer un ARMA(1,3): 
estimate p=1 q=3;run; 
 
Par défaut, la procédure ARIMA utilise les moindres carrés conditionnels pour estimer les 
paramètres. Si on veut utiliser la méthode du maximum de vraisemblance : 
estimate p=2 method=ml;run; 
ou la méthode des moindres carrés non conditionnels: 
estimate p=2 method=uls;run; 
 
Rajouter l’option plot permet d’avoir la représentation des autocorrélations simples, inverses 
et partielles des résidus du modèle estimé. 
 
Pour obtenir les résidus estimés (pour faire ensuite des tests de validation) et l’ajustement de 
la série par le modèle que l’on vient d’estimer, il faut passer à la commande forecast : 
forecast out=a.res lead=0;run; 
 
L’option lead=n spécifie l’horizon auquel on construit la prévision. Ici lead=0 signifie 
qu’on calcule l’ajustement et non la prévision. 
 
Dans la table a.res, il y a 6 variables : lconso (série intiale), forecast (ajustement de la 
série si on a mis lead=0), std (l’écart type de la prévision), L95 (limite inférieur de 
l’intervalle de confiance à 95%), U95 (limite supérieure de l’intervalle de confiance à 95%) et 
residual (les résidus estimés). 
 
Une autre option de la commande forecast est back=n qui spécifie le nombre n 
d’observations avant la fin de l’échantillon à partir duquel les prévisions doivent commencer. 
Par défaut back=0. 


